Uždavinys „Gilės“.
Jaunesniųjų sprendimas
Nepaisant sudėtingos sąlygos, šis uždavinys labai paprastas. Programai pakanka simuliuoti voverės judesius ir kiekvienu žingsniu patikrinti, ar einamuoju momentu einamojoje pozicijoje pagaunama gilė. Norint tai patikrinti efektyviai, galima skaitant pradinius duomenis užpildyti dvimatį loginį, arba skaičių masyvą:

const MAXT = 10000;

      MAXP = 15;

var giles : array [1..MAXT, 1..MAXP] of integer;
Programos veikimo pradžioje šis masyvas užpildytas nuliais. Perskaičius duomenis apie pozicijoje pk laiko momentu tk nukrentančią gilę, giles[tk, pk] reikšmę padidinsime vienetu.
Alfos atliekamus ėjimus (šokinėjimus) patogu saugoti skaičių masyve: -1 žymėsime ėjimą į kairę, 0 – stovėjimą vietoje, o 1 – ėjimą į dešinę. Mūsų programoje šie ėjimai saugomi masyve ejimai:

var ejimai : array [1..MAXT] of integer;
Tuomet uždavinį sprendžia šis paprastas algoritmas:

pradiniai_duomenys;     { užpildomi masyvai “giles” ir “ejimai”, 

                          kintamieji T ir P }
pozicija  := (P + 1) div 2;

atsakymas := 0;

for laikas := 1 to T do begin
    inc(pozicija, ejimai[laikas]);

    inc(atsakymas, giles[laikas, pozicija]);
end;

rezultatai(atsakymas);  { atsakymas įrašomas į rezultatų failą }
Algoritmo sudėtingumas yra 

, kadangi atliekamų elementarių veiksmų skaičius tiesiškai priklauso nuo T. Programai taip pat reikia perskaityti duomenis apie N gilių iš failo. Taigi bendras programos veikimo sudėtingumas yra 

, jis visiškai pakankamas visiems testams įveikti per leistiną laiką.
Jei duomenis apie giles saugotume N ilgio sąraše ir kiekvienu žingsniu šį sąrašą „perbėgtume“, tikrindami, ar jokia gilė pozicijoje p nenukrenta laiko momentu t, algoritmo sudėtingumas būtų 

, nepakankamas įveikti didesniesiems testams. Tačiau ir tokią paiešką galima pagreitinti, kadangi duomenys apie giles išrikiuoti pagal laiko momentus.
Pilna uždavinį sprendžianti programa pateikta šio dokumento pabaigoje.

Vyresniųjų sprendimas

Vyresniųjų uždavinys yra gerokai sudėtingesnis (ir įdomesnis!): prašoma nustatyti, kiek daugiausiai gilių gali pagauti Alfa.

Nėra jokios paprastos strategijos (taisyklių rinkinio), kurios laikydamasi Alfa spėtų sugauti maksimalų skaičių gilių. Šį uždavinį išspręsime dinaminiu programavimu.

Tegu g(t, p) yra maksimalus gilių, sugaunamų iš pozicijos p nuo laiko momento t, skaičius. Mus dominantis atsakymas yra g(0, (P + 1) / 2). g(t, p) visada lygu 0, kai t didesnis už bet kurios gilės nukritimo momentą (toliau jį žymėsime T ), t.y. kai jokios gilės nebesulauksime.

Tarkime, voverė yra pozicijoje p laiko momentu t. Sekančiu (t + 1-uoju) laiko momentu ji galės atsidurti tik vienoje iš p - 1-osios, p-osios ir p + 1-osios pozicijų (išskyrus tuos atvejus, kai p yra viena iš kraštinių pozicijų). Taigi g(t, p) reikšmė gali būti lygi tik vienai iš reikšmių 
g(t + 1, p - 1), g(t + 1, p) ir g(t + 1, p + 1), plius vienas, jei laiko momentu t pozicijoje p taip pat nukrenta gilė. Kadangi g(t, p) yra maksimalus gilių skaičius, tai, žinoma, visuomet reikia pridėti didžiausią iš minėtų reikšmių. Patogumo dėlei taip pat apibrėšime g(t, p) = 0, jei p nepriklauso intervalui 1..P. Taigi:

 lygu laiko momentu t pozicijoje p nukrentančių gilių skaičiui (0 arba 1).
Naudodami šį apibrėžimą, galėtume g(0, (P + 1) / 2) reiškmę suskaičiuoti rekursyviai, tačiau tai būtų neefektyvu, nes tos pačios funkcijos reikšmės būtų skaičiuojamos labai daug kartų (nesunku įsitikinti, kad atliekamų veiksmų skaičius augtų eksponentiškai didėjant T ).
Vienas problemos sprendimo būdas – skaičiuoti funkcijos reikšmes rekursyviai, tačiau kartą suskaičiuotą reikšmę įsiminti T × P dydžio lentelėje ir jos neperskaičiuoti antrą kartą. Galima apsieiti be rekursijos, visą funkcijos g reikšmių lentelę užpildant „nuo viršaus iki apačios“. Mes savo programoje šią lentelę saugosime masyve g:
const MAXP = 15;

      MAXT = 10000;

var g : array [0..MAXT, 0..MAXP + 1] of longint;


Programos veikimo pradžioje šis masyvas užpildytas nuliais. Perskaičius duomenis apie pozicijoje pk laiko momentu tk nukrentančią gilę, g[tk, pk] reikšmę padidinsime vienetu (Pav. 2). Pabaigus skaityti pradinius duomenis, nuo viršaus iki apačios paeiliui išnagrinėsime visus lentelės langelius: langelio [t, p] reikšmę padidinsime didžiausiu skaičiumi iš [t + 1, p - 1], [t + 1, p] ir [t + 1, p + 1] (Pav. 3). Aptartą sprendimo būdą atitinka šis algoritmas:

pradiniai_duomenys;    
{ gilių nukritimo vietos pažymimos 1,

  suskaičiuojamas maks. laiko momentas T }
for laikas := T - 1 downto 0 do
    for pozicija := 1 to P do
        inc(g[laikas, pozicija], 
            max(g[laikas + 1, pozicija - 1],
                g[laikas + 1, pozicija], 
                g[laikas + 1, pozicija + 1]));

{ atsakymas – reikšmė g[0, (P + 1) div 2] }

rezultatai(g[0, (P + 1) div 2]);

Šio algoritmo sudėtingumas 

 (čia T – didžiausias gilės nukritimo laiko momentas) – visiškai pakankamas visiems testams įveikti per leistiną laiką.
Pilna uždavinį sprendžianti programa pateikta šio dokumento pabaigoje.

Kitas sprendimo būdas

Galimas ir kitas neblogas sprendimo būdas, kuris taip pat remiasi dinaminiu programavimu. Gilę apibūdina jos nukritimo momentas, bei pozicija, kurioje ji nukrenta. Tegu (t1, p1) ir (t2, p2) yra dvi gilės. Pagavus pirmąją gilę, antrą Alfa galės pagauti tik tada, jei | p2 - p1| ≤ t2 - t1. Tegu mi – didžiausias skaičius gilių, kurias galime būti pagavę prieš pagaudami i-tąją gilę (įskaitant ir ją pačią). Tuomet:

Ir uždavinį sprendžia toks algoritmas:

pradiniai_duomenys;

atsakymas := 0;

for i := 1 to N do begin
    m[i] := 0;

    for j := 1 to i - 1 do
        if abs(p[i] - p[j]) <= t[i] - t[j] then
            if m[i] < m[j] then m[i] := m[j];

    inc(m[i]);

    if m[i] > atsakymas then atsakymas := m[i];

end;

rezultatai(atsakymas);

Tačiau kad jis veiktų korektiškai, skaitant pradinius duomenis turi būti „išmetamos“ tos gilės, kurių Alfa tikrai nespės pagauti (sąlyga | (P + 1) / 2 – pk | > tk).
Šio algoritmo sudėtingumas yra 

, pakankamas įveikti testams, kur N ≤ 5000.

Testai

Tie patys testai buvo pateikiami tiek vyresniųjų, tiek ir jaunesniųjų programoms. 

Kadangi jaunesniųjų „Gilių“ sprendimas labai paprastas, labiau buvo vertinamos programos, teisingai veikiančios su didesniais testais: paskutiniai keturi testai buvo vertinami po 10 taškų, o pirmieji 12 – po 5 taškus.

Sudarinėjant testus, daugiau dėmesio buvo skirta vyresniųjų sprendimų teisingumui vertinti. Visus testus turėtų įveikti tik teisingas ir efektyvus, dinaminiu programavimu pagrįstas algoritmas. Daugiau negu pusę taškų turėtų gauti neefektyvios, bet teisingai veikiančios programos. Mažiau – godžiuoju algoritmu arba euristika pagrįstos programos. Atsakymus spėliojančios programos neturėtų įveikti daugiau negu tris testus.
	Nr.
	Pozicijų sk. ( P )
	Didžiausias laiko momentas ( T )
	Gilių sk. ( N )
	Atsakymas
	Komentaras

	0 a
	5
	5
	6
	3
	Pavyzdiniai testai.

	0 b
	7
	5
	7
	4
	

	1
	3
	3
	3
	2
	Nedideli testai.

	2
	5
	5
	6
	5
	

	3
	5
	7
	3
	2
	

	4
	1
	15
	8
	8
	

	5
	7
	7
	6
	5
	

	6
	3
	10
	10
	6
	

	7
	5
	10
	9
	6
	

	8
	7
	9
	11
	5
	

	9
	9
	30
	42
	15
	Vidutiniai testai.

	10
	13
	100
	125
	65
	

	11
	15
	200
	550
	80
	

	12
	15
	450
	2 000
	350
	

	13
	7
	4 000
	4 800
	2 400
	Didesnieji testai.

	14
	13
	5 000
	20 200
	4 998
	

	15
	15
	10 000
	40 000
	8 198
	

	16
	15
	10 000
	49 800
	9 801
	


Programa jgiles.pas
program JGiles;

    const MAXT = 10000;

          MAXP = 15;

    var giles   : array [1..MAXT, 1..MAXP] of integer;

        ejimai  : array [1..MAXT] of integer;

        P, T, N : longint;

procedure pradiniai_duomenys;

var f : text;

    raide : char;

    tk, pk, k : longint;

begin
    assign(f, 'GILES.IN'); reset(f);

    readln(f, P, T, N);

    for k := 1 to T do begin
        read(f, raide);

        case raide of
            'K' : ejimai[k] := -1;

            'D' : ejimai[k] :=  1;

            else  ejimai[k] :=  0;

        end;

    end;

    readln(f);

    for k := 1 to N do begin
        readln(f, tk, pk);

        inc(giles[tk, pk]);

    end;

    close(f);

end;

procedure rezultatai(ats : longint);

var f : text;

begin
    assign(f, 'GILES.OUT'); rewrite(f);

    writeln(f, ats);

    close(f);

end;

    var laikas, pozicija, atsakymas : longint;

begin
    pradiniai_duomenys;     { užpildomi masyvai “giles” ir “ejimai”, 

                              kintamieji T ir P }

    pozicija  := (P + 1) div 2;

    atsakymas := 0;

    for laikas := 1 to T do begin
        inc(pozicija, ejimai[laikas]);

        inc(atsakymas, giles[laikas, pozicija]);

    end;

    rezultatai(atsakymas);  { atsakymas įrašomas į rezultatų failą }
end.

Programa vgiles.pas
program VGiles;

    const MAXP = 15;

          MAXT = 10000;

    var g : array [0..MAXT, 0..MAXP + 1] of longint;

        P, N, T : longint;

procedure pradiniai_duomenys;

var f : text;

    tk, pk, k : longint;

begin
    assign(f, 'GILES.IN'); reset(f);

    readln(f, P);

    readln(f, N);

    for k := 1 to N do
    begin
        readln(f, tk, pk);

        inc(g[tk, pk]);

        if tk > T then T := tk;

    end;

    close(f);

end;

procedure rezultatai(ats : longint);

var f : text;

begin
    assign(f, 'GILES.OUT'); rewrite(f);

    writeln(f, ats);

    close(f);

end;

function max(a, b, c : longint) : longint;

begin
    if (a > b) and (a > c) then max := a

    else if b > c then max := b

    else max := c

end;

    var laikas, pozicija : longint;

begin
    pradiniai_duomenys;    { gilių nukritimo vietos pažymimos 1,

                             suskaičiuojamas maks. laiko momentas T }
    for laikas := T – 1 downto 0 do
        for pozicija := 1 to P do
            inc(g[laikas, pozicija],

                max(g[laikas + 1, pozicija - 1],

                    g[laikas + 1, pozicija],

                    g[laikas + 1, pozicija + 1]));

                           { atsakymas – reikšmė g[0, (P + 1) div 2] }
    rezultatai(g[0, (P + 1) div 2]);

end.

Pav. � SEQ Pav. \* ARABIC �1� Pradiniai duomenys.





Pav. � SEQ Pav. \* ARABIC �2� Lentelės užpildymas nuskaičius pradinius duomenis.





Pav. � SEQ Pav. \* ARABIC �3� Prie langelio [t, p] pridedama didžiausia iš langelių [t + 1, p - 1], [t + 1, p] ir [t + 1, p + 1] reikšmių





Pav. � SEQ Pav. \* ARABIC �4� Atlikus skaičiavimus, atsakymas langelyje [0, (P+1) div 2].

